
モデル理論アプローチによるシミュレーション

Moore型オートマトン多層ネットワークのモデル
東洋大学　旭貴朗

付録　多層ネットワークS6のCAST言語によるモデル

注意

　「CAST言語によるシミュレーション」サイト[注1]で公開されているsimcast.zip（＋extProlog）をインストールすると，モデル理論アプローチによるシミュレーション開発実行環境simcastができあがり，本文中のモデルを実行することができる．開発実行環境の操作方法などは，同サイトや高原康彦ほか『形式手法 モデル理論アプローチ：情報システム開発の基礎』日科技連(2007)（少なくとも第３章3.1項pp.87-103）を参照されたい．
１　はじめに

　本稿の目的は，モデル理論アプローチによるシミュレーション開発実行環境simcast [注1] を使用して，複数のMoore型オートマトンを要素とするネットワークを多層的に組合わせた複雑なシステムのシミュレーションを行なう方法を紹介することである．

　すでに，状態機械を要素とする多層ネットワークのモデル化技法について論じてある [注2] ので，ここでは，Moore型オートマトンに対象を限定し[注3]，複雑なシステムの段階的な開発が可能であることを報告する．使用する言語はCASTである[注4]．

　以下，第２節では，要素システムであるMoore型オートマトンを解説する．第３節では，３つの要素が結合するネットワークについて考察し，単層ネットワーク（同期型固定結合網）の一般構造を抽出する．また，CAST言語による単層ネットワークのモデル化の方法を提案する．第４節では，複数の単層ネットワークが多層的にネットワークを構成するときのモデルを例示する．第５節は結語である．

　　　　　　　　表紙の「注意」を挿入
２　要素システム

　ここでは，要素システムであるMoore型オートマトンを解説し，そのCAST言語によるモデルを紹介する．

2.1 Moore型オートマトン

　Moore型オートマトンは，入力に応じて状態が遷移し，遷移後の状態に対して出力が決まるようなシステムのモデルである．

　　　[image: image2.wmf]　　　　　　[image: image3.wmf]　
　　図１　要素システムS1　　　　　　　　図２　状態遷移図

　ある時刻でのシステムSへの入力をa，出力をb，状態をcとする（図１）．また，次の時刻での状態をccと書くことにする（ccは２つのcを乗じたものではなく，ひとつの記号として扱う）．これらの変数の間に，

δ(c,a)=ccかつλ(c)=b

の関係があるとき，Sはオートマトンであるといい，<δ,λ, c0 >をS1のMoore型オートマトンモデルと呼ぶ．ただし，δを状態遷移関数，λを出力関数， c0を初期状態と呼ぶ．

　状態遷移図は図２のようになる．　時刻tにおける状態がcのとき，出力はλ(c) である．そこにaが入力されると，次の時刻t+1における状態はcc=δ(c,a) となり，b'=λ(δ(c,a)) が出力される．

　一般に，出力がその時刻の入力のみに依存して定まるようなシステムを静的システムと呼び，それ以外を動的システムと呼ぶ．この言い方からすると，現在の出力が過去の入力と過去の状態によって定まるオートマトンは動的システムである．

2.2 Moore型オートマトンのCAST言語によるモデル

　具体的な例として，入力をそのまま出力するが，その入力を状態として一時的に保持するようなシステムS1を考える．これをMoore型オートマトンとしてモデル化すると，

δ1(c,a) =cc <-> cc=a

λ1(c) = b <-> b=c

となる．ただし，初期状態をc0=0とする．このモデル <δ1,λ1, 0 > は，aが入力されたとき，次の状態ccがaと等しくなり，出力bもaと等しくなるようなモデルである．結果として，S1のオートマトンモデルとなっている．

　さて，上記のオートマトンモデルをCAST言語で書くと図４のようになる．テキストエディタでこれを作成し，例えばsim10.setというファイル名で保存する．ここで，開発環境MTA-SDKを起動し，ファイルsim10.setをコンパイルすると自動的に実行され，図５の最終３行のような結果が得られる．


//sim10.set　要素システムS1のモデル

func([delta1,lambda1]);

delta1(c,a)=cc <-> cc:=a;

lambda1(c)=b <-> b:=c;

inputsequence()=[1,0,0,0,0,0,0,0,0,0,0,0,0];

initialstate()=0;

delta(c,a)=delta1(c,a);

lambda(c)=lambda1(c);

　図４　要素システムS1のCAST言語によるモデル

図５　モデル（図４）のコンパイルと実行

　下から３行目のinputsequenceの右辺の中のnilは，初期入力（時刻０での入力）がないことを表している．nilの下に，初期状態0と初期出力0が示されている．

　２項目の１が時刻１での入力である．その1の下に，それに対する遷移後の状態が１，出力が１であることが示されている．

　オートマトンを実行するエンジンstdAutomatonEngine.pの動作について述べる．まず，initialstate()から初期状態を取り出し，lambda(c)を使って，初期出力を計算する．次に，入力系列inputsequence()から順次入力を取り出して，delta(c,a)とlambda(c)を繰り返して計算する．本質的に，エンジンは，delta(c,a)とlambda(c)の右辺で定義されているものを，交互に，実行する（計算する）だけのものである．

３　単層ネットワーク

　単層ネットワーク（同期型固定結合網）の一般的構造とCAST言語によるモデルの実装構造，および例として循環するネットワークのモデルを示す．

3.1 循環するネットワークのモデル

　単層ネットワークの例として，図６のような循環するネットワークを考える．

[image: image5.wmf]
　　　　　　　　　　　　　図６　循環するネットワークS4

　このシステムS4は，３つの動的システム（要素システムS1,S2,S3）と静的システム（加算要素(+)）を含む，ネットワークシステムである．次の図７は，図６の配置を換え，動的システムS1, S2, S3が並列に並ぶように描き直したものである．このように配置すると，要素システムの結合が図の左側にひとつにまとまり，後述するような一貫した理解ができるようになる． 
[image: image6.wmf]
　　　　　　　　　　図７　循環するネットワークS4の構造

　外部からの入力aと出力bb4と要素システムへの入力aa=(a1,a2,a3) の間の関係を入力結合と呼ぶ．それを関数として表現するなら，aa=in(a,bb4)と書くことができる．循環するネットワークS4の例では，(a1,a2,a3)=(a+b3,b1,b2)に対応する．また，出力bb4と外部への出力の関係は，b=out(bb4)と書くことができる．S4の例では，b=b2となっている．ただし，出力bb4=(b1,b2,b3) は，次の時刻にどれかの要素システムへの入力として使用されるので，入力側に帰還（フィードバック）している．
　これらを使えば，つぎのように全体システムの状態遷移関数と出力関数を定義することができる．
δ4(c,a)=cc <-> 
bb4=(λ1(c1),λ2(c2),λ1(c3) ),

(a1,a2,a3)=in(a,bb4), 

cc=(δ1(c1,a1),δ1(c1,a1),δ1(c1,a1) ).

λ4(c)=b <-> 

bb4=(λ1(c1),λ2(c2),λ1(c3) ),

b=out(bb4).

　したがって，S4はMoore型オートマトンであり，<δ4,λ4,(0,0,0) >がそのオートマトンモデルである．これをCAST言語でモデル化すると，次のようになる（図８）． 


//sim40.set　循環するネットワーク

func([delta1,lambda1,delta2,lambda2,delta3,lambda3,delta4,lambda4]);

delta1(c,a)=cc <-> cc:=a;

lambda1(c)=b <-> b:=c;

delta2(c,a)=delta1(c,a);

lambda2(c)=lambda1(c);

delta3(c,a)=delta1(c,a);

lambda3(c)=lambda1(c);

delta4(c,a)= cc <->

[c1,c2,c3]:=c,


[b1,b2,b3]:=bb4.g,

//前時刻の出力を取得

[a1,a2,a3]:=[a+b3,b1,b2],
//入力結合

cc:=[delta1(c1,a1),delta2(c2,a2),delta3(c3,a3)];   //次の状態

lambda4(c)= b <->

[c1,c2,c3]:=c,

bb4:=[lambda1(c1),lambda2(c2),lambda3(c3)],

bb4.g:=bb4,


//この時刻の出力を保存

b:=project(bb4,2);

//出力結合

inputsequence()=[1,0,0,0,0,0,0,0,0,0,0,0,0];

initialstate()=[0,0,0];

delta(c,a)=delta4(c,a);

lambda(c)=lambda4(c);

　図８　ネットワークS4のCAST言語によるモデル

　ただし，オートマトンを実行するエンジンstdAutomatonEngine.pは２変数関数delta(c,a)と１変数関数lambda(c)を交互に実行するが，lambda4(c)を先に計算するので，bb4=(λ1(c1),λ2(c2),λ1(c3) ) はすでに計算されている．これをグローバル変数bb4.gに代入しておき，あとでdelta4(c,a)を計算するときに再利用している．

　また，このネットワークは３つの要素をもつシステムなので，状態はc=(c1,c2,c3)である．そのため，初期状態initialstate()は３項組 [0,0,0] となっている． 

　要素システムが円状に結合しているので，入力結合は [a1,a2,a3]:=[a+b3,b1,b2] である．これが式１の入力結合(a1,a2,a3)=in(a,bb4)の具体的な形である．外部からの入力aと内部要素の出力(b1,b2,b3)がここで混合され，各要素Siへの入力aiとなる．

　出力結合は b:=project(bb4,2) となっている．これが式１の b=out(bb) の具体的な形である．bb4:=[b1,b2,b3] の中からb2を抽出している．実行結果は図９に示した．

　図９　モデル（図８）の実行結果
3.2 単層ネットワークの実装構造

　一般に，どのように結合した単層ネットワーク全体でも，式(1)のように，各要素システムの状態の組を状態とする，ひとつのMoore型オートマトンとしてモデル化することができる．したがって，単層ネットワークのCAST言語によるモデルの一般的な形は図９のようになる．


delta(c,a)= cc <->


//f(c,a)=ccのこと

[c1,c2,c3]:=c,


[b1,b2,b3]:=bb.g,

//前時刻の出力を取得

[a1,a2,a3]:=in(a,b1,b2,b3),
//in( )は入力結合

cc:=[delta1(c1,a1),delta2(c2,a2),delta3(c3,a3)];   //次の状態

lambda(c)= b <->


//g(c)=bのこと

[c1,c2,c3]:=c,

bb:=[lambda1(c1),lambda2(c2),lambda3(c3)],

bb.g:=bb,


//この時刻の出力を保存

b:=out(bb);


//out( )は出力結合

　　　図10　CAST言語による単層ネットワークの実装構造

　図10は３つの要素をもつネットワークのモデルであり，テンプレートとして使用することができる．要素の数が変わればdeltaiとlambdaiを増減させ，入力結合in()や出力結合out()を状況に応じて変えれば良い．例えば，前項（図８）のシステムS4では，入力結合in(a,bb)は，[a1,a2,a3]:=[a+b3,b1,b2] ，出力結合out(bb)は，b:=project(bb4,2) であった．

　ただし，静的システムによる自己フィードバックについては前稿「CAST言語による状態機械の多層ネットワークモデル」を参照されたい．また，状況に応じて，結合そのものが変化したり，同じ入力に同時に複数の入力があるようなシステムの場合は，出力記憶bb.gを，{["b1" v1],["b2" v2],["b3" v3]}のように，出力名とその値の組の集合で表記し，それを操作するような入力結合関数in( ) を作成すればよい [注5]．

４　多層ネットワーク

　前節で紹介した方法で，多層ネットワークをモデル化する．例として，次のネットワークS6をCAST言語でモデル化する（図11）．

[image: image8.wmf]
　　　　　　　　　　　　図11　多層ネットワークS6

　システムS6は，単層ネットワークS4とひとつのオートマトンS6が結合した多層ネットワークである．しかし，いまやS4もひとつのオートマトンであるとみなせるので，結果として，システムS6を２つのオートマトンが直列結合している単層ネットワークとみなすことができる．したがって，前節の単層ネットワークを実装構造（図10）と同じように，モデル化することができる（図12）．実行結果は図13に示した．


〜〜〜〜〜　中略　〜〜〜〜〜〜

delta6(c,a)= cc <->

[c4,c5]:=c,

[b4,b5]:=bb6.g,

[a4,a5]:=[a,b4],    //in(a,bb6)に相当

cc:=[delta4(c4,a4),delta5(c5,a5)]; 

lambda6(c)= b <->

[c4,c5]:=c,

bb6:=[lambda4(c4),lambda5(c5)],


bb6.g:=bb6,

b:=project(bb6,2);
inputsequence()=[1,0,0,0,0,0,0,0,0,0,0,0,0];

initialstate()=[[0,0,0],0];

delta(c,a)=delta6(c,a);

lambda(c)=lambda6(c);

　図12　多層ネットワークS6のCAST言語によるモデルの一部（詳細は付録２）
　このネットワークは２つの要素をもつシステムだが，S4の状態がc=(c1,c2,c3)で，S5の状態がc5であるため，全体としての状態は，c=((c1,c2,c3),c5)となる．そのため，初期状態initialstate()は [[0,0,0],0] となっている． 

　要素システムが直列に結合しているので，入力結合in(a,bb6)は [a4,a5]:=[a,b4] である． 出力結合out(bb6)は b:=project(bb6,2) となっている．実行結果は図１４に示した．

　　　　　図13　モデル（図12）の実行結果

５　おわりに

　モデル理論アプローチによるシステム開発環境MTA-SDKにおいて，単層ネットワークを積み重ねて，段階的に多層ネットワーク（同期型固定結合網）をモデル化することができることが判明した． 

　段階的モデル化における設計技法の主要点をまとめると，

１）単層ネットワークに対しては，要素の動的システム（オートマトン）を並列に並べた図7を考え，図10のようにモデル化する（第３節）．

２）その時，すべての要素の出力をグローバル変数bb.gに代入したり，参照したりする．

３）多層ネットワークに対しては，要素がネットワークであっても，それをひとつの動的システム（オートマトン）とみなして，単層ネットワークと同様にモデル化する（第４節）．

４）以上を繰り返して，複雑な多層ネットワークのモデルを構成していく．

となる．

　この方法でモデル化すれば，記述も単純になり，一貫した視点で多層ネットワークのモデルを構成していくことが可能である．
注

[1] 開発実行環境simcastは，公開サイト「CAST言語によるシミュレーション」
http://www2.toyo.ac.jp/~asahi/research/simulation/ からダウンロードできる．

[2] 旭貴朗「CAST言語による状態機械の多層ネットワークモデル」技術資料
[3]「書法として」Moore型オートマトンに限定する主旨である．Mealey型オートマトンはMoore型オートマトンに形式変換することができるので，実質的にはシミュレーション対象が制限されているとは言えない．

[4] 公開サイト「モデル理論アプローチによる情報システム開発」

http://www.geocities.jp/kisoten/

のダウンロードページからCASTの言語仕様が入手できる．また，旭貴朗ほか「経営情報システム開発のためのモデル記述言語：CAST」経営情報学会誌 (2008掲載予定) も参照可能である．

[5] 例えば，h(x,bb)=y <-> member([x,y],bb); と定義しておけば，入力結合in(a,bb.g)=aa <-> aa:=[h(bb.g,"a1"), h(bb.g,"a2"),h(bb.g,"a3")]; とbb.g:=[["a1",5],["a2",6],["a3",7]]に対して，aa=[a1,a2,a3]=[5,6,7]が得られる．関数hを工夫して，複数入力の設定も可能である．
付録（多層ネットワークS6のCAST言語によるモデル）
//sim60.set
func([delta1,lambda1,delta2,lambda2,delta3,lambda3,

delta4,lambda4,delta5,lambda5,delta6,lambda6]);

delta1(c,a)=cc <-> cc:=a;

lambda1(c)=b <-> b:=c;

delta2(c,a)=delta1(c,a);

lambda2(c)=lambda1(c);

delta3(c,a)=delta1(c,a);

lambda3(c)=lambda1(c);

delta4(c,a)= cc <->

[c1,c2,c3]:=c,

[b1,b2,b3]:=bb4.g,


[a1,a2,a3]:=[a+b3,b1,b2],

cc:=[delta1(c1,a1),delta2(c2,a2),delta3(c3,a3)];

lambda4(c)= b <->

[c1,c2,c3]:=c,

bb4:=[lambda1(c1),lambda2(c2),lambda3(c3)],

bb4.g:=bb4,

b:=project(bb4,2);

delta5(c,a)=delta1(c,a);

lambda5(c)=lambda1(c);

delta6(c,a)= cc <->

[c4,c5]:=c,

[b4,b5]:=bb6.g,

[a4,a5]:=[a,b4],

cc:=[delta4(c4,a4),delta5(c5,a5)];

lambda6(c)= b <->

[c4,c5]:=c,

bb6:=[lambda4(c4),lambda5(c5)],

bb6.g:=bb6,

b:=project(bb6,2);

inputsequence()=[1,0,0,0,0,0,0,0,0,0,0,0,0];

initialstate()=[[0,0,0],0];

delta(c,a)=delta6(c,a);

lambda(c)=lambda6(c);

